March 26, 2024.

Phugoid continues : linearization at a fixed point.

Note: No class on Thursday, March 28.

project-related stuff.

```
>
```

Returning to the phugoid:

```
> R:='R':  # just in case it got changed.
  phug :=[D(theta)(t)=v(t)- cos(theta(t))/v(t), D(v)(t)=-sin(theta(t))-R*v(t)^2 ];
  xphug:=[op(phug), D(x)(t)=v(t)*cos(theta(t)), D(y)(t)=v(t)*sin(theta(t))]:
```

$$phug := \left[D(\theta)(t) = v(t) - \frac{\cos(\theta(t))}{v(t)}, D(v)(t) = -\sin(\theta(t)) - R\,v(t)^2\right] \quad (1)$$

From last time, let's write a function that given a value of R, solves the equation for the fixed point.

```
> FP:=proc(Rval)
   local mypair,eqns;
   mypair:=
    convert(
      solve(
       map(eq->rhs(eq)=0,
         subs({theta(t)=theta, v(t)=v}, phug)),
      {v,theta}),
     radical);
   eqns:=eval(mypair,R=Rval);
   return(subs(eqns,[theta,v]));
  end:
> FP(R);
```

$$\left[\arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), \left(\frac{1}{R^2+1}\right)^{1/4}\right] \quad (2)$$

```
> FP(0.25);
```

$$[-0.2449786631, 0.9849581210] \quad (3)$$

I want to look at how the fixed point depends on R

need to transform $\left[\arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), \left(\frac{1}{R^2+1}\right)^{1/4}\right]$ *into a parametric version*

```
> [op(FP(R)), R=0..3]
```

$$\left[\arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), \left(\frac{1}{R^2+1}\right)^{1/4}, R = 0..3\right] \quad (4)$$

> $plot\left([op(FP(R)), R=0..5], \theta = -\frac{\mathrm{Pi}}{2}..\frac{\mathrm{Pi}}{2}, v = 0..1.5, title = \text{"fixed point as a function of R"}, size = [.6, .4]\right)$



Want to study the linearization at the fixed point as a function of R. But there is work to do

```
> phug
```

$$\left[D(\theta)(t) = v(t) - \frac{\cos(\theta(t))}{v(t)}, D(v)(t) = -\sin(\theta(t)) - R\,v(t)^2\right] \quad (5)$$

```
> map(rhs, phug);
```

$$\left[v(t) - \frac{\cos(\theta(t))}{v(t)}, -\sin(\theta(t)) - R\,v(t)^2\right] \quad (6)$$

```
> subs({v(t) = v, theta(t) = theta}, map(rhs, phug));
```

$$\left[v - \frac{\cos(\theta)}{v}, -\sin(\theta) - R\,v^2\right] \quad (7)$$

```
> F:=proc(x,y,Rv:='R')
    local vf;
   vf:=subs({v(t)=y,theta(t)=x}, map(rhs,phug));
   return( eval(eval(vf, R=Rv)));
  end
```

(8)

```
F := proc(x, y, Rv := 'R')
    local vf;
    vf := subs({θ(t) = x, v(t) = y}, map(rhs, phug)); return eval(eval(vf, R = Rv))
end proc
```

```
> vf;
```

$$vf \quad (9)$$

```
> F(v, theta)
```

$$\left[\theta - \frac{\cos(v)}{\theta}, -\sin(v) - R\,\theta^2\right] \quad (10)$$

```
> F(.2, Pi/6)
```

$$\left[-1.348190509, -0.1986693308 - \frac{R\,\pi^2}{36}\right] \quad (11)$$

```
> F(.2, 1/2, RRR)
```

$$\left[-1.460133156, -0.1986693308 - \frac{RRR}{4}\right] \quad (12)$$

```
> F(.2, .7, .3)
```

$$[-0.700095112, -0.3456693308] \quad (13)$$

> $F(0, 1)$

$$[0, -R] \quad (14)$$

> $with(VectorCalculus):$

> $Jack := Jacobian(F(\theta, v), [\theta, v]);$

$$Jack := \begin{bmatrix} \frac{\sin(\theta)}{v} & 1 + \frac{\cos(\theta)}{v^2} \\ -\cos(\theta) & -2\,R\,v \end{bmatrix} \quad (15)$$

This is good, but want to evaluate Jack at the fixed point. Recall what we've done

> $FP(.2);$

$$[-0.1973955598, 0.9902427357] \quad (16)$$

If R=0.2, there is a fixed point at -0.1973..., 0.9902... Jacoian there is

>

> $eval(Jack, \{\theta = -0.1973955598, v = 0.9902427357, R = 0.2\})$

$$\begin{bmatrix} -0.1980485471 & 2.000000000 \\ -0.9805806757 & -0.3960970942 \end{bmatrix} \quad (17)$$

> $Eigenvalues(\%);$

$$Eigenvalues\left(\begin{bmatrix} -0.1980485471 & 2.000000000 \\ -0.9805806757 & -0.3960970942 \end{bmatrix}\right) \quad (18)$$

> $with(LinearAlgebra):$

> $Eigenvalues(eval(Jack, \{\theta = -0.1973955598, v = 0.9902427357, R = 0.2\}))$

$$\begin{bmatrix} -0.297072820650000 + 1.39690928289846\,I \\ -0.297072820650000 - 1.39690928289846\,I \end{bmatrix} \quad (19)$$

> $Eigenvalues(eval(Jack, \{\theta = 0, v = 1, R = 0\}))$

$$\begin{bmatrix} I\sqrt{2} \\ -I\sqrt{2} \end{bmatrix} \quad (20)$$

Want to plug in FP(R) for theta and v, then compute Jack, then get eigenvalues there.

```
> GetEVatFP:=proc(Rval)
    local fix, Jac, theta, v,ev;
    fix:=FP(Rval);
    Jac:=Jacobian(F(theta,v),[theta,v]);
    ev:=Eigenvalues(eval(Jac, {theta=fix[1],v=fix[2]}));
    return(eval(ev,R=Rval));
  end:
```

> $GetEVatFP(.2);$

$$\begin{bmatrix} -0.2970728206 + 1.396909283\,I \\ -0.2970728206 - 1.396909283\,I \end{bmatrix} \quad (21)$$

> $GetEVatFP(3);$

$$\left[\begin{array}{c} -\frac{2\,10^{3/4}}{5} \\ -\frac{10^{3/4}}{2} \end{array}\right] \quad \textbf{(22)}$$

> $GetEVatFP(0);$

$$\left[\begin{array}{c} \sqrt{-2} \\ -\sqrt{-2} \end{array}\right] \quad \textbf{(23)}$$

> $GetEVatFP(MMM);$

$$\left[\begin{array}{c} -\frac{3\left(\frac{1}{MMM^2+1}\right)^{1/4} MMM}{2} + \frac{\sqrt{MMM^2\sqrt{\frac{1}{MMM^2+1}} - 8\sqrt{\frac{1}{MMM^2+1}}}}{2} \\ -\frac{3\left(\frac{1}{MMM^2+1}\right)^{1/4} MMM}{2} - \frac{\sqrt{MMM^2\sqrt{\frac{1}{MMM^2+1}} - 8\sqrt{\frac{1}{MMM^2+1}}}}{2} \end{array}\right] \quad \textbf{(24)}$$

>